

Interpolazione di Lagrange e interpolazione di Hermite

nome del file: lagrange_hermite.mws
autore: Claudio Marsan
ultima modifica: 23 gennaio 2002
testato con: Maple V release 6.02 su Windows 2000

```
> restart;  
> with(plots):  
Warning, the name changecoords has been redefined
```

Funzione per calcolare il polinomio di interpolazione di Lagrange:

```
> Lagrange := proc(x, ax, ay)  
  local value, i, n, j, prod;  
  value := 0;  
  n := nops(ax) - 1;  
  for i from 0 to n do  
    prod := 1;  
    for j from 0 to n do  
      if j <> i then  
        prod := prod*(x-ax[j+1])/(ax[i+1]-ax[j+1]);  
      end if;  
    end do;  
    value := value + ay[i+1] *prod;  
  end do;  
end:
```

Il vettore delle ascisse note:

```
> X := [0, Pi/2, Pi];
```

$$X := \left[0, \frac{1}{2}\pi, \pi \right]$$

Il vettore delle ordinate note:

```
> Y := map(sin, X);
```

$$Y := [0, 1, 0]$$

Il vettore delle "velocità" note:

```
> dY := map(D(sin), X);
```

$$dY := [1, 0, -1]$$

Il polinomio di interpolazione di Hermite che bisogna costruire:

```
> p := x -> a[0] + sum(a[j]*x^j, j = 1.. 5):
```

```
> p(x);
```

$$a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5$$

La derivata del polinomio di Hermite:

```
> dp := D(p):
```

```
> dp(x);
```

$$a_1 + 2 a_2 x + 3 a_3 x^2 + 4 a_4 x^3 + 5 a_5 x^4$$

Costruzione del sistema lineare di 6 equazioni in 6 incognite:

```
> for k from 1 to 3 do eq[2*k - 1] := p(X[k]) = Y[k]; end do:
```

```
[ > for k from 1 to 3 do eq[2*k] := dp(X[k]) = dY[k]; end do:
> for i from 1 to 6 do eq[i]; end do;
      a0 = 0
      a1 = 1
      a0 +  $\frac{1}{2} a_1 \pi + \frac{1}{4} a_2 \pi^2 + \frac{1}{8} a_3 \pi^3 + \frac{1}{16} a_4 \pi^4 + \frac{1}{32} a_5 \pi^5 = 1$ 
      a1 + a2 π +  $\frac{3}{4} a_3 \pi^2 + \frac{1}{2} a_4 \pi^3 + \frac{5}{16} a_5 \pi^4 = 0$ 
      a0 + a1 π + a2 π2 + a3 π3 + a4 π4 + a5 π5 = 0
      a1 + 2 a2 π + 3 a3 π2 + 4 a4 π3 + 5 a5 π4 = -1
```

[Risoluzione del sistema:

```
[ > A := solve({eq[1], eq[2], eq[3], eq[4], eq[5], eq[6]}, {a[0],
a[1], a[2], a[3], a[4], a[5]}):
> assign(A):
> for i from 0 to 5 do a[i]; end do;
```

$$\begin{array}{c}
0 \\
1 \\
-\frac{5\pi - 16}{\pi^2} \\
8\frac{-4 + \pi}{\pi^3} \\
-4\frac{-4 + \pi}{\pi^4} \\
0
\end{array}$$

[Il polinomio di interpolazione di Hermite:

```
[ > p(x);
      x -  $\frac{(5\pi - 16)x^2}{\pi^2} + \frac{8(-4 + \pi)x^3}{\pi^3} - \frac{4(-4 + \pi)x^4}{\pi^4}$ 
> sort(evalf(p(x)));
      .03524957831 x4 - .2214796326 x3 + .02958950714 x2 + x
```


[Il polinomio di interpolazione di Lagrange:

```
[ > Lagrange(x, X, Y);
      -4  $\frac{x(x - \pi)}{\pi^2}$ 
> sort(expand(evalf(%)));
      -.4052847344 x2 + 1.273239544 x
```

[Confronto tra i grafici di sin(x) [nero], il grafico del polinomio interpolatore di Hermite [rosso] e il grafico del polinomio interpolatore di Lagrange [blu] sull'intervallo [0, Pi]:

```
[ > gr1 := plot(sin(x), x = 0.. Pi, color = black):
> gr2 := plot(p(x), x = 0.. Pi, color = red):
```

```
[ > gr3 := plot(Lagrange(x, X, Y), x = 0.. Pi, color = blue):  
> display([gr1, gr2, gr3]);
```



Confronto tra l'errore commesso mediante l'interpolazione di Hermite [rosso] e mediante l'interpolazione di Lagrange [blu] approssimando $\sin(x)$ sull'intervallo $[0, \pi]$:

```
[ > gr4 := plot(p(x) - sin(x), x = 0.. Pi, color = red):  
> gr5 := plot(Lagrange(x, X, Y) - sin(x), x = 0.. Pi, color =  
blue):  
> display([gr4, gr5]);
```

